Pensieve header: The full sl_2 invariant using the Drinfel'd double. Continues 2018-05/ybax.nb, Talks/StonyBrook-1805/ybax.nb, Projects/SL2Portfolio/Logoi.nb.

Profiling

In[∘]:=
```
Once[
  SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\SL2Invariant"];
  << KnotTheory`;
  << "../Profile/Profile.m";
 ];
BeginProfile[];
Once@PopupWindow[Button["Show Profile Monitor"],
  Dynamic[PrintProfile[], UpdateInterval → 3, TrackedSymbols → {}]]
```

```
Loading KnotTheory` version of January 20, 2015, 10:42:19.1122.
Read more at http://katlas.org/wiki/KnotTheory.
This is Profile.m of http://www.drorbn.net/AcademicPensieve/Projects/Profile/.
This version: June 2018. Original version: July 1994.
```

Out[∘]= Show Profile Monitor

External Utilities

In[∘]:=
```
HL[ℰ_] := Style[ℰ, Background → Yellow];
```

Program

Program

Internal Utilities

Program

Canonical Form:

Program

In[∘]:=
```
CCF[ℰ_] := PP_CCF@ExpandDenominator@ExpandNumerator@PP_Together@Together[PP_Exp[
      Expand[ℰ] //. e^x_ e^y_ :→ e^(x+y) /. e^x_ :→ e^CCF[x]]];
CF[ℰ_List] := CF /@ ℰ;
CF[sd_SeriesData] := MapAt[CF, sd, 3];
CF[ℰ_] := PP_CF@Module[
    {vs = Cases[ℰ, (y | b | t | a | x | η | β | τ | α | ξ)_, ∞] ⋃ {y, b, t, a, x, η, β, τ, α, ξ}},
    Total[CoefficientRules[Expand[ℰ], vs] /. (ps_ → c_) :→ CCF[c] (Times @@ vs^ps)]
   ];
```

Program

The Kronecker δ:

Program

In[∘]:=
```
Kδ /: Kδ_{i_,j_} := If[i === j, 1, 0];
```

Program

Equality, multiplication, and degree-adjustment of perturbed Gaussians; $\mathbb{E}[L, Q, P]$ stands for $e^{L+Q} P$:

Program

In[∘]:=
```
𝔼 /: 𝔼[L1_, Q1_, P1_] ≡ 𝔼[L2_, Q2_, P2_] :=
  CF[L1 == L2] ∧ CF[Q1 == Q2] ∧ CF[Normal[P1 - P2] == 0];
𝔼 /: 𝔼[L1_, Q1_, P1_] 𝔼[L2_, Q2_, P2_] := 𝔼[L1 + L2, Q1 + Q2, P1 * P2];
𝔼[L_, Q_, P_]_{$k_} := 𝔼[L, Q, Series[Normal@P, {ϵ, 0, $k}]];
```

Program

Zip and Bind

Program

Variables and their duals:

Program

In[∘]:=
```
{t*, b*, y*, a*, x*, z*} = {τ, β, η, α, ξ, ζ};
{τ*, β*, η*, α*, ξ*, ζ*} = {t, b, y, a, x, z}; (u_{i_})* := (u*)_i;
```

Program

Finite Zips:

Program

In[∘]:=
```
collect[sd_SeriesData, ζ_] := MapAt[collect[#, ζ] &, sd, 3];
collect[ℰ_, ζ_] := PP_Collect@Collect[ℰ, ζ];
Zip_{}[P_] := P; Zip_{ζ_,ζs___}[P_] := PP_Zip[
  (collect[P // Zip_{ζs}, ζ] /. f_. ζ^d_. :> ∂_{ζ*,d} f) /. ζ* → 0]
```

Program

QZip implements the “Q-level zips” on $\mathbb{E}(L, Q, P) = \mathrm{P}e^{L+Q}$. Such zips regard the L variables as scalars.

Program

In[∘]:=
```
QZip_{ζs_List}@𝔼[L_, Q_, P_] := PP_QZip@Module[{ζ, z, zs, c, ys, ηs, qt, zrule, Q1, Q2},
    zs = Table[ζ*, {ζ, ζs}];
    c = CF[Q /. Alternatives @@ (ζs ⋃ zs) → 0];
    ys = CF@Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
    ηs = CF@Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
    qt = CF@Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ}Q, {ζ, ζs}, {z, zs}];
    zrule = Thread[zs → CF[qt.(zs + ys)]];
    Q2 = CF[(Q1 = CF[c + ηs.zs /. zrule]) /. Alternatives @@ zs → 0];
    CF /@ 𝔼[L, Q2, Det[qt] e^{-Q2} Zip_{ζs}[e^{Q1} (P /. zrule)]]];
```

In[∘]:=
```
$T = 1;
QZip_{ζs_List}@𝔼[L_, Q_, P_] :=
  PP_QZip@Module[{ζ, z, zs, c, ys, ηs, qt, zrule, Q1, Q2, tin = TimeUsed[], out},
    zs = Table[ζ*, {ζ, ζs}];
    c = CF[Q /. Alternatives @@ (ζs ⋃ zs) → 0];
    ys = CF@Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
    ηs = CF@Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
    qt = CF@Inverse@Table[Kδ_{z,ζ*} − ∂_{z,ζ}Q, {ζ, ζs}, {z, zs}];
    zrule = Thread[zs → CF[qt.(zs + ys)]];
    Q2 = CF[(Q1 = CF[c + ηs.zs /. zrule]) /. Alternatives @@ zs → 0];
    out = CF /@ 𝔼[L, Q2, Det[qt] 𝕖^{-Q2} Zip_{ζs}[𝕖^{Q1} (P /. zrule)]];
    (*If[(tin=TimeUsed[]-tin)>$T, $T=2tin; Echo[{tin,ζs,L,Q,P}]];*) out ];
```

Program

Upper to lower and lower to Upper:

Program

In[∘]:=
```
U2l = {B_i^{p_·} → 𝕖^{-p ℏ γ b_i}, B^{p_·} → 𝕖^{-p ℏ γ b}, T_i^{p_·} → 𝕖^{p ℏ t_i}, T^{p_·} → 𝕖^{p ℏ t}, 𝒜_i^{p_·} → 𝕖^{p γ α_i}, 𝒜^{p_·} → 𝕖^{p γ α}};
l2U = {𝕖^{c_· b_i + d_·} :→ B_i^{-c/(ℏ γ)} 𝕖^d, 𝕖^{c_· b + d_·} :→ B^{-c/(ℏ γ)} 𝕖^d,
    𝕖^{c_· t_i + d_·} :→ T_i^{c/ℏ} 𝕖^d, 𝕖^{c_· t + d_·} :→ T^{c/ℏ} 𝕖^d,
    𝕖^{c_· α_i + d_·} :→ 𝒜_i^{c/γ} 𝕖^d, 𝕖^{c_· α + d_·} :→ 𝒜^{c/γ} 𝕖^d,
    𝕖^{ℰ_} :→ 𝕖^{Expand@ℰ}};
```

Program

LZip implements the "L-level zips" on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard all of Pe^Q as a single"P". Here the z's are b and α and the ζ's are β and a.

Program

In[∘]:=
```
LZip_{ζs_List}@𝔼[L_, Q_, P_] := PP_LZip@Module[{ζ, z, zs, c, ys, ηs, lt, zrule, L1, L2, Q1, Q2},
    zs = Table[ζ*, {ζ, ζs}];
    c = L /. Alternatives @@ (ζs ⋃ zs) → 0;
    ys = Table[∂_ζ (L /. Alternatives @@ zs → 0), {ζ, ζs}];
    ηs = Table[∂_z (L /. Alternatives @@ ζs → 0), {z, zs}];
    lt = Inverse@Table[Kδ_{z,ζ*} − ∂_{z,ζ}L, {ζ, ζs}, {z, zs}];
    zrule = Thread[zs → lt.(zs + ys)];
    L2 = (L1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
    Q2 = (Q1 = Q /. U2l /. zrule) /. Alternatives @@ zs → 0;
    CF /@ (CF /@ 𝔼[L2, Q2, Det[lt] 𝕖^{-L2-Q2} Zip_{ζs}[𝕖^{L1+Q1} (P /. U2l /. zrule)]] //. l2U) ];
```

Program

In[∘]:=
```
B_{}[L_, R_] := L R;
B_{is__}[L_𝔼, R_𝔼] := PP_B@Module[{n},
    Times[
      L /. Table[(v : b | B | t | T | a | x | y)_i → v_{n@i}, {i, {is}}],
      R /. Table[(v : β | τ | α | 𝒜 | ξ | η)_i → v_{n@i}, {i, {is}}]
    ] // LZip_{Join@@Table[{β_{n@i}, τ_{n@i}, a_{n@i}}, {i, {is}}]} // QZip_{Join@@Table[{ξ_{n@i}, y_{n@i}}, {i, {is}}]} ];
B_{is___}[L_, R_] := B_{{is}}[L, R];
```

Program

$\mathbb{E}$ morphisms with domain and range.

Program

In[∘]:=
```
B_is_List[𝔼_d1_→r1_[L1_, Q1_, P1_], 𝔼_d2_→r2_[L2_, Q2_, P2_]] :=
  𝔼_(d1⋃Complement[d2,is])→(r2⋃Complement[r1,is]) @@ B_is[𝔼[L1, Q1, P1], 𝔼[L2, Q2, P2]];
𝔼_d1_→r1_[L1_, Q1_, P1_] // 𝔼_d2_→r2_[L2_, Q2_, P2_] :=
  B_r1⋂d2[𝔼_d1→r1[L1, Q1, P1], 𝔼_d2→r2[L2, Q2, P2]];
𝔼_d1_→r1_[L1_, Q1_, P1_] ≡ 𝔼_d2_→r2_[L2_, Q2_, P2_] ^:=
  (d1 == d2) ∧ (r1 == r2) ∧ (𝔼[L1, Q1, P1] ≡ 𝔼[L2, Q2, P2]);
𝔼_d1_→r1_[L1_, Q1_, P1_] 𝔼_d2_→r2_[L2_, Q2_, P2_] ^:=
  𝔼_(d1⋃d2)→(r1⋃r2) @@ (𝔼[L1, Q1, P1] 𝔼[L2, Q2, P2]);
𝔼_d_→r_[L_, Q_, P_]_$k_ := 𝔼_d→r @@ 𝔼[L, Q, P]_$k;
𝔼_[ℰ___][i_] := {ℰ}⟦i⟧;
```

Program

“Define” code

Program

Define[lhs = rhs,   ...] defines the lhs to be rhs, except that rhs is computed only once for each value of $k. Fancy Mathematica not for the faint of heart. Most readers should ignore.

Program

In[∘]:=
```
SetAttributes[Define, HoldAll];
Define[def_, defs__] := (Define[def]; Define[defs];);
Define[op_is__ = ℰ_] := Module[{SD, ii, jj, kk, isp, nis, nisp, sis}, Block[{i, j, k},
    ReleaseHold[Hold[
        SD[op_nisp,$k_Integer, PP_Boot@Block[{i, j, k}, op_isp,$k = ℰ; op_nis,$k]];
        SD[op_isp, op_{is},$k]; SD[op_sis__, op_{sis}];
      ] /. {SD → SetDelayed,
        isp → {is} /. {i → i_, j → j_, k → k_},
        nis → {is} /. {i → ii, j → jj, k → kk},
        nisp → {is} /. {i → ii_, j → jj_, k → kk_}
      }] ]]
```

Program

Booting Up

Program

In[∘]:=
```
$k = 2; (*ℏ=γ=1;*)
```

Program

In[∘]:=
```
Define[am_i,j→k = 𝔼_{i,j}→{k}[(α_i + α_j) a_k, (ⅇ^(-γ α_j) ξ_i + ξ_j) x_k, 1]_$k,
  bm_i,j→k = 𝔼_{i,j}→{k}[(β_i + β_j) b_k, (η_i + η_j) y_k, ⅇ^((ⅇ^(-ϵ β_i) - 1) η_j y_k)]_$k]
```

Program

In[∘]:=

```
Define[R_{i,j} = E_{{}→{i,j}}[ħ a_j b_i, ħ x_j y_i, e^(Sum_{k=2}^{$k+1} ((1 - e^{γ ϵ ħ})^k (ħ y_i x_j)^k) / (k (1 - e^{k γ ϵ ħ})))]_{$k},
  R̄_{i,j} = E_{{}→{i,j}}[-ħ a_j b_i, -ħ x_j y_i / B_i, 1 + If[$k == 0, 0, (R̄_{{i,j},$k-1})_{$k}[3] -
      (((R̄_{{i,j},0})_{$k} R_{1,2} (R̄_{{3,4},$k-1})_{$k}) // (bm_{i,1→i} am_{j,2→j}) // (bm_{i,3→i} am_{j,4→j}))[3]]],
  P_{i,j} = E_{{i,j}→{}}[β_i α_j / ħ, η_i ξ_j / ħ, 1 + If[$k == 0, 0, (P_{{i,j},$k-1})_{$k}[3] -
      (R_{1,2} // ((P_{{1,j},0})_{$k} (P_{{i,2},$k-1})_{$k}))[3]]]]
```

Program

In[∘]:=

```
Define[aS_j = R̄_{i,j} ~ B_i ~ P_{i,j},
  aS̄_i = E_{{i}→{i}}[-a_i α_i, -x_i A_i ξ_i, 1 + If[$k == 0, 0, (aS̄_{{i},$k-1})_{$k}[3] -
      ((aS̄_{{i},0})_{$k} ~ B_i ~ aS_i ~ B_i ~ (aS̄_{{i},$k-1})_{$k})[3]]]]
```

Program

In[∘]:=

```
Define[bS_i = R_{i,1} ~ B_1 ~ aS_1 ~ B_1 ~ P_{i,1},
  bS̄_i = R_{i,1} ~ B_1 ~ aS̄_1 ~ B_1 ~ P_{i,1},
  aΔ_{i→j,k} = (R_{1,j} R_{2,k}) // bm_{1,2→3} // P_{3,i},
  bΔ_{i→j,k} = (R_{j,1} R_{k,2}) // am_{1,2→3} // P_{i,3}]
```

Program

In[∘]:=

```
Define[dm_{i,j→k} = (E_{{i,j}→{i,j}}[β_i b_i + α_j a_j, η_i y_i + ξ_j x_j, 1]
      (aΔ_{i→1,2} // aΔ_{2→2,3} // aS̄_3) (bΔ_{j→-1,-2} // bΔ_{-2→-2,-3})) // (P_{-1,3} P_{-3,1} am_{2,j→k} bm_{i,-2→k}),
  dS_i = E_{{i}→{1,2}}[β_i b_1 + α_i a_2, η_i y_1 + ξ_i x_2, 1] // (bS̄_1 aS_2) // dm_{2,1→i},
  dΔ_{i→j,k} = (bΔ_{i→3,1} aΔ_{i→2,4}) // (dm_{3,4→k} dm_{1,2→j})]
```

Program

In[∘]:=

```
Define[C_i = E_{{}→{i}}[0, 0, B_i^{1/2} e^{-ħ ϵ a_i/2}]_{$k},
  C̄_i = E_{{}→{i}}[0, 0, B_i^{-1/2} e^{ħ ϵ a_i/2}]_{$k},
  Kink_i = (R_{1,3} C̄_2) // dm_{1,2→1} // dm_{1,3→i},
  K̄ink_i = (R̄_{1,3} C_2) // dm_{1,2→1} // dm_{1,3→i}]
```

Program

Note. $t == \epsilon \mathrm{a} - \gamma \mathrm{b}$ and $b == -t/\gamma + \epsilon \mathrm{a}/\gamma$.

Program

In[∘]:=

```
Define[b2t_i = E_{{i}→{i}}[α_i a_i - β_i t_i / γ, ξ_i x_i + η_i y_i, e^{ϵ β_i a_i/γ}]_{$k},
  t2b_i = E_{{i}→{i}}[α_i a_i - τ_i γ b_i, ξ_i x_i + η_i y_i, e^{ϵ τ_i a_i}]_{$k}]
```

Testing

In[∘]:=
```
Block[{$k = 1}, {
   am → am_{i,j→k}, bm → bm_{i,j→k}, dm → dm_{i,j→k}, R → R_{i,j}, R̄ → R̄_{i,j}, P → P_{i,j},
   aS → aS_i, aS̄ → aS̄_i, bS → bS_i, bS̄ → bS̄_i, dS → dS_i, aΔ → aΔ_{i→j,k}, bΔ → bΔ_{i→j,k},
   dΔ → dΔ_{i→j,k}, C → C_i, C̄ → C̄_i, Kink → Kink_i, Kink̄ → Kink̄_i, b2t → b2t_i, t2b → t2b_i
  }] //
 Column
```

Out[∘]=

$$am \to \mathbb{E}_{\{i,j\}\to\{k\}}\left[a_k\ (\alpha_i+\alpha_j),\ x_k\left(e^{-\gamma\alpha_j}\xi_i+\xi_j\right),\ 1\right]$$

$$bm \to \mathbb{E}_{\{i,j\}\to\{k\}}\left[b_k\ (\beta_i+\beta_j),\ y_k\ (\eta_i+\eta_j),\ 1-y_k\beta_i\eta_j\epsilon+O[\epsilon]^2\right]$$

$$dm \to \mathbb{E}_{\{i,j\}\to\{k\}}\Bigg[a_k\alpha_i+a_k\alpha_j+b_k\beta_i+b_k\beta_j,\ y_k\eta_i+\frac{y_k\eta_j}{\mathcal{A}_i}+\frac{x_k\xi_i}{\mathcal{A}_j}+\frac{(1-B_k)\eta_j\xi_i}{\hbar}+x_k\xi_j,$$
$$1+\left(-\frac{y_k\beta_i\eta_j}{\mathcal{A}_i}-\frac{x_k\beta_j\xi_i}{\mathcal{A}_j}+a_kB_k\eta_j\xi_i+\frac{\gamma\hbar x_ky_k\eta_j\xi_i}{\mathcal{A}_i\mathcal{A}_j}+\frac{(\gamma-3\gamma B_k)y_k\eta_j^2\xi_i}{2\mathcal{A}_i}+\frac{(\gamma-3\gamma B_k)x_k\eta_j\xi_i^2}{2\mathcal{A}_j}+\frac{(\gamma-4\gamma B_k+3\gamma B_k^2)\eta_j^2\xi_i^2}{4\hbar}\right)\epsilon+$$
$$O[\epsilon]^2\Bigg]$$

$$R \to \mathbb{E}_{\{\}\to\{i,j\}}\left[\hbar a_jb_i,\ \hbar x_jy_i,\ 1-\frac{1}{4}\left(\gamma\hbar^3x_j^2y_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{R} \to \mathbb{E}_{\{\}\to\{i,j\}}\left[-\hbar a_jb_i,\ -\frac{\hbar x_jy_i}{B_i},\ 1+\left(-\frac{\hbar^2a_jx_jy_i}{B_i}-\frac{3\gamma\hbar^3x_j^2y_i^2}{4B_i^2}\right)\epsilon+O[\epsilon]^2\right]$$

$$P \to \mathbb{E}_{\{i,j\}\to\{\}}\left[\frac{\alpha_j\beta_i}{\hbar},\ \frac{\eta_i\xi_j}{\hbar},\ 1+\frac{\gamma\eta_i^2\xi_j^2\epsilon}{4\hbar}+O[\epsilon]^2\right]$$

$$aS \to \mathbb{E}_{\{i\}\to\{i\}}\left[-a_i\alpha_i,\ -x_i\mathcal{A}_i\xi_i,\ 1+\left(-\hbar a_ix_i\mathcal{A}_i\xi_i-\frac{1}{2}\gamma\hbar x_i^2\mathcal{A}_i^2\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{aS} \to \mathbb{E}_{\{i\}\to\{i\}}\left[-a_i\alpha_i,\ -x_i\mathcal{A}_i\xi_i,\ 1+\left(\gamma\hbar x_i\mathcal{A}_i\xi_i-\hbar a_ix_i\mathcal{A}_i\xi_i-\frac{1}{2}\gamma\hbar x_i^2\mathcal{A}_i^2\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$bS \to \mathbb{E}_{\{i\}\to\{i\}}\left[-b_i\beta_i,\ -\frac{y_i\eta_i}{B_i},\ 1+\left(-\frac{y_i\beta_i\eta_i}{B_i}-\frac{\gamma\hbar y_i^2\eta_i^2}{2B_i^2}\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{bS} \to \mathbb{E}_{\{i\}\to\{i\}}\left[-b_i\beta_i,\ -\frac{y_i\eta_i}{B_i},\ 1+\left(\frac{\gamma\hbar y_i\eta_i}{B_i}-\frac{y_i\beta_i\eta_i}{B_i}-\frac{\gamma\hbar y_i^2\eta_i^2}{2B_i^2}\right)\epsilon+O[\epsilon]^2\right]$$

$$dS \to \mathbb{E}_{\{i\}\to\{i\}}\Bigg[-a_i\alpha_i-b_i\beta_i,\ -\frac{y_i\mathcal{A}_i\eta_i}{B_i}-x_i\mathcal{A}_i\xi_i+\frac{(\mathcal{A}_i-B_i\mathcal{A}_i)\eta_i\xi_i}{\hbar B_i},$$
$$1+\Bigg(\frac{\gamma\hbar y_i\mathcal{A}_i\eta_i}{B_i}-\frac{y_i\mathcal{A}_i\beta_i\eta_i}{B_i}-\frac{\gamma\hbar y_i^2\mathcal{A}_i^2\eta_i^2}{2B_i^2}-\hbar a_ix_i\mathcal{A}_i\xi_i-x_i\mathcal{A}_i\beta_i\xi_i+\frac{a_i\mathcal{A}_i\eta_i\xi_i}{B_i}-$$
$$\frac{\gamma\hbar x_iy_i\mathcal{A}_i^2\eta_i\xi_i}{B_i}+\frac{(-\gamma\mathcal{A}_i+\gamma B_i\mathcal{A}_i)\eta_i\xi_i}{B_i}+\frac{(\mathcal{A}_i-B_i\mathcal{A}_i)\beta_i\eta_i\xi_i}{\hbar B_i}+\frac{y_i\left(3\gamma\mathcal{A}_i^2-\gamma B_i\mathcal{A}_i^2\right)\eta_i^2\xi_i}{2B_i^2}-$$
$$\frac{1}{2}\gamma\hbar x_i^2\mathcal{A}_i^2\xi_i^2+\frac{x_i\left(3\gamma\mathcal{A}_i^2-\gamma B_i\mathcal{A}_i^2\right)\eta_i\xi_i^2}{2B_i}+\frac{\left(-3\gamma\mathcal{A}_i^2+4\gamma B_i\mathcal{A}_i^2-\gamma B_i^2\mathcal{A}_i^2\right)\eta_i^2\xi_i^2}{4\hbar B_i^2}\Bigg)\epsilon+O[\epsilon]^2\Bigg]$$

$$a\Delta \to \mathbb{E}_{\{i\}\to\{j,k\}}\left[a_j\alpha_i+a_k\alpha_i,\ x_j\xi_i+x_k\xi_i,\ 1+\left(-\hbar a_jx_k\xi_i+\frac{1}{2}\gamma\hbar x_jx_k\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$b\Delta \to \mathbb{E}_{\{i\}\to\{j,k\}}\left[b_j\beta_i+b_k\beta_i,\ B_ky_j\eta_i+y_k\eta_i,\ 1+\frac{1}{2}\gamma\hbar B_ky_jy_k\eta_i^2\epsilon+O[\epsilon]^2\right]$$

$$d\Delta \to \mathbb{E}_{\{i\}\to\{j,k\}}\Bigg[a_j\alpha_i+a_k\alpha_i+b_j\beta_i+b_k\beta_i,\ y_j\eta_i+B_jy_k\eta_i+x_j\xi_i+x_k\xi_i,$$
$$1+\left(\frac{1}{2}\gamma\hbar B_jy_jy_k\eta_i^2-\hbar a_jx_k\xi_i+\frac{1}{2}\gamma\hbar x_jx_k\xi_i^2\right)\epsilon+O[\epsilon]^2\Bigg]$$

$$C \to \mathbb{E}_{\{\}\to\{i\}}\left[0,\ 0,\ \sqrt{B_i}-\frac{1}{2}\left(\hbar a_i\sqrt{B_i}\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{C} \to \mathbb{E}_{\{\}\to\{i\}}\left[0,\ 0,\ \frac{1}{\sqrt{B_i}}+\frac{\hbar a_i\epsilon}{2\sqrt{B_i}}+O[\epsilon]^2\right]$$

$$Kink \to \mathbb{E}_{\{\}\to\{i\}}\left[\hbar a_ib_i,\ \hbar x_iy_i,\ \frac{1}{\sqrt{B_i}}+\left(\frac{\hbar a_i}{2\sqrt{B_i}}-\frac{\gamma\hbar^3x_i^2y_i^2}{4\sqrt{B_i}}\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{Kink} \to \mathbb{E}_{\{\}\to\{i\}}\left[-\hbar a_ib_i,\ -\frac{\hbar x_iy_i}{B_i},\ \sqrt{B_i}+\left(-\frac{1}{2}\hbar a_i\sqrt{B_i}-\frac{\hbar^2a_ix_iy_i}{\sqrt{B_i}}-\frac{3\gamma\hbar^3x_i^2y_i^2}{4B_i^{3/2}}\right)\epsilon+O[\epsilon]^2\right]$$

$$b2t \to \mathbb{E}_{\{i\}\to\{i\}}\left[a_i\alpha_i-\frac{t_i\beta_i}{\gamma},\ y_i\eta_i+x_i\xi_i,\ 1+\frac{a_i\beta_i\epsilon}{\gamma}+O[\epsilon]^2\right]$$

$$t2b \to \mathbb{E}_{\{i\}\to\{i\}}\left[a_i\alpha_i-\gamma b_i\tau_i,\ y_i\eta_i+x_i\xi_i,\ 1+a_i\tau_i\epsilon+O[\epsilon]^2\right]$$

Check that on the generators this agrees with our conventions in the handout:

```
In[•]:= Timing@
  {{"[a,x]" → ((𝔼{}→{1,2}[0, 0, a2 x1] // am1,2→1) [3] - (𝔼{}→{1,2}[0, 0, a1 x2] // am1,2→1) [3]),
     "[b,y]" → ((𝔼{}→{1,2}[0, 0, y2 b1] // bm1,2→1) [3] - (𝔼{}→{1,2}[0, 0, y1 b2] // bm1,2→1) [3])} /.
    z_1 → z,
   {"Δ[y]" → Last[𝔼{}→{1}[0, 0, y1] ~B1~ bΔ1→1,2],
    "Δ[b]" → Last[𝔼{}→{1}[0, 0, b1] ~B1~ bΔ1→1,2],
    "Δ[a]" → Last[𝔼{}→{1}[0, 0, a1] ~B1~ aΔ1→1,2],
    "Δ[x]" → Last[𝔼{}→{1}[0, 0, x1] ~B1~ aΔ1→1,2]},
   {
     "S(a)" → ((𝔼{}→{1}[0, 0, a1] ~B1~ aS1) [3]),
     "S(x)" → ((𝔼{}→{1}[0, 0, x1] ~B1~ aS1) [3]),
     "S(b)" → ((𝔼{}→{1}[0, 0, b1] ~B1~ bS1) [3]),
     "S(y)" → ((𝔼{}→{1}[0, 0, y1] ~B1~ bS1) [3])
   } /. z_1 → z}
```

Out[•]= $\{1.67188, \{\{[a,x] \to -x\gamma,\ [b,y] \to -y\epsilon + O[\epsilon]^3\},\ \{\triangle[y] \to (B_2 y_1 + y_2) + O[\epsilon]^3,\ \triangle[b] \to (b_1+b_2) + O[\epsilon]^3,$
$\triangle[a] \to (a_1+a_2) + O[\epsilon]^3,\ \triangle[x] \to (x_1+x_2) - \hbar a_1 x_2 \epsilon + \frac{1}{2}\hbar^2 a_1^2 x_2 \epsilon^2 + O[\epsilon]^3\},\ \{S(a) \to -a + O[\epsilon]^3,$
$S(x) \to -x - a x \hbar \epsilon - \frac{1}{2}(a^2 x \hbar^2)\epsilon^2 + O[\epsilon]^3,\ S(b) \to -b + O[\epsilon]^3,\ S(y) \to -\frac{y}{B} + O[\epsilon]^3\}\}\}$

Hopf algebra axioms on both sides separately.

Associativity of am and bm:

```
In[•]:= Timing@Block[{$k = 3},
   HL /@ {(am1,2→1 // am1,3→1) ≡ (am2,3→2 // am1,2→1), (bm1,2→1 // bm1,3→1) ≡ (bm2,3→2 // bm1,2→1)}
  ]
```

Out[•]= {0.28125, {True, True}}

R and *P* are inverses:

```
In[•]:= Timing@Block[{$k = 3}, {Ri,j, Pi,k, HL[(Ri,j // Pi,k) ≡ 𝔼{k}→{j}[aj αk, xj ξk, 1]]}]
```

Out[•]= $\{0.203125, \{\mathbb{E}_{\{\}\to\{i,j\}}[\hbar a_j b_i, \hbar x_j y_i, 1 - \frac{1}{4}(\gamma \hbar^3 x_j^2 y_i^2)\epsilon + (\frac{1}{9}\gamma^2\hbar^5 x_j^3 y_i^3 + \frac{1}{32}\gamma^2\hbar^6 x_j^4 y_i^4)\epsilon^2 +$
$\frac{1}{1152}(24\gamma^3\hbar^5 x_j^2 y_i^2 - 72\gamma^3\hbar^7 x_j^4 y_i^4 - 32\gamma^3\hbar^8 x_j^5 y_i^5 - 3\gamma^3\hbar^9 x_j^6 y_i^6)\epsilon^3 + O[\epsilon]^4],$
$\mathbb{E}_{\{i,k\}\to\{\}}[\frac{\alpha_k \beta_i}{\hbar}, \frac{\eta_i \xi_k}{\hbar}, 1 + \frac{\gamma \eta_i^2 \xi_k^2 \epsilon}{4\hbar} + \frac{1}{288\hbar^2}(36\gamma^2\hbar^2\eta_i^2\xi_k^2 + 40\gamma^2\hbar\eta_i^3\xi_k^3 + 9\gamma^2\eta_i^4\xi_k^4)\epsilon^2 +$
$(\frac{1}{24}\gamma^3\hbar\eta_i^2\xi_k^2 + \frac{1}{6}\gamma^3\eta_i^3\xi_k^3 + \frac{13\gamma^3\eta_i^4\xi_k^4}{96\hbar} + \frac{5\gamma^3\eta_i^5\xi_k^5}{144\hbar^2} + \frac{\gamma^3\eta_i^6\xi_k^6}{384\hbar^3})\epsilon^3 + O[\epsilon]^4], \text{True}\}\}$

as and $\overline{\text{aS}}$ are inverses, bs and $\overline{\text{bS}}$ are inverses:

```
In[•]:= Timing[HL /@ {(aS̅1 // aS1) ≡ 𝔼{1}→{1}[a1 α1, x1 ξ1, 1], (bS̅1 // bS1) ≡ 𝔼{1}→{1}[b1 β1, y1 η1, 1]}]
```

Out[•]= {0.578125, {True, True}}

(co)-associativity on both sides

In[∘]:=
```
Timing[
 HL /@ {(aΔ_{1→1,2} // aΔ_{2→2,3}) ≡ (aΔ_{1→1,3} // aΔ_{1→1,2}), (bΔ_{1→1,2} // bΔ_{2→2,3}) ≡ (bΔ_{1→1,3} // bΔ_{1→1,2}),
   (am_{1,2→1} // am_{1,3→1}) ≡ (am_{2,3→2} // am_{1,2→1}), (bm_{1,2→1} // bm_{1,3→1}) ≡ (bm_{2,3→2} // bm_{1,2→1})}]
```
Out[∘]= {0.71875, {True, True, True, True}}

Δ is an algebra morphism

In[∘]:=
```
Timing[HL /@ {(am_{1,2→1} // aΔ_{1→1,2}) ≡ ((aΔ_{1→1,3} aΔ_{2→2,4}) // (am_{3,4→2} am_{1,2→1})),
   (bm_{1,2→1} // bΔ_{1→1,2}) ≡ ((bΔ_{1→1,3} bΔ_{2→2,4}) // (bm_{3,4→2} bm_{1,2→1}))}]
```
Out[∘]= {0.9375, {True, True}}

An explicit formula for aS_i

In[∘]:=
```
Timing@Block[{$k = 4}, HL[aS_i ≡ (𝔼_{{i}→{i,j}}[-α_i a_j, -ξ_i x_i,
      Sum[Expand[(e^{ξ_i x_i} (-ℏ γ ϵ)^k)/(2^k k!) Nest[Expand[x_i^2 ∂_{{x_i,2}} #] &, e^{-ξ_i e^{ℏ ϵ a_i} x_i}, k]], {k, 0, $k}]]_{$k} //
    am_{i,j→i})]]
```
Out[∘]= {4.6875, True}

S is convolution inverse of id

In[∘]:=
```
Timing[HL[# ≡ 𝔼_{{1}→{1}}[0, 0, 1]] & /@ {
   (aΔ_{1→1,2} ~ B_1 ~ aS_1) ~ B_{1,2} ~ am_{1,2→1}, (aΔ_{1→1,2} ~ B_2 ~ aS_2) ~ B_{1,2} ~ am_{1,2→1},
   (bΔ_{1→1,2} ~ B_1 ~ bS_1) ~ B_{1,2} ~ bm_{1,2→1}, (bΔ_{1→1,2} ~ B_2 ~ bS_2) ~ B_{1,2} ~ bm_{1,2→1}}]
```
Out[∘]= {0.875, {True, True, True, True}}

But not with the opposite product:

In[∘]:=
```
Timing[Short[# ≡ 𝔼_{{1}→{1}}[0, 0, 1]] & /@ {
   (aΔ_{1→1,2} ~ B_1 ~ aS_1) ~ B_{1,2} ~ am_{2,1→1}, (aΔ_{1→1,2} ~ B_2 ~ aS_2) ~ B_{1,2} ~ am_{2,1→1},
   (bΔ_{1→1,2} ~ B_1 ~ bS_1) ~ B_{1,2} ~ bm_{2,1→1}, (bΔ_{1→1,2} ~ B_2 ~ bS_2) ~ B_{1,2} ~ bm_{2,1→1}}]
```
Out[∘]= $\{1., \{\frac{1}{2}\left(-2\gamma\epsilon\hbar x_1 \mathcal{A}_1 \xi_1 + \gamma^2\epsilon^2\hbar^2 x_1 \mathcal{A}_1 \xi_1 - 2\gamma\epsilon^2\hbar^2 \ll 1 \gg x_1 \mathcal{A}_1 \xi_1 + 2\gamma^2\epsilon^2\hbar^2 x_1^2 \mathcal{A}_1^2 \xi_1^2\right) == 0,$

$\frac{1}{2}\left(-2\gamma\epsilon\hbar x_1 \xi_1 - \gamma^2\epsilon^2\hbar^2 x_1 \xi_1 + 2\gamma^2\epsilon^2\hbar^2 x_1^2 \xi_1^2\right) == 0,$

$\frac{1}{2}\left(-2\gamma\epsilon\hbar y_1 \eta_1 - \gamma^2\epsilon^2\hbar^2 y_1 \eta_1 + 2\gamma^2\epsilon^2\hbar^2 y_1^2 \eta_1^2\right) == 0,$

$\frac{1}{2 B_1^2}\left(-2\gamma\epsilon\hbar B_1 y_1 \eta_1 + \ll 1 \gg - \ll 1 \gg + 2\gamma^2\epsilon^2\hbar^2 y_1^2 \eta_1^2\right) == 0\}\}$

S is an algebra anti-(co)morphism

In[∘]:=
```
Timing[HL /@ {am_{1,2→1} ~ B_1 ~ aS_1 ≡ (aS_1 aS_2) ~ B_{1,2} ~ am_{2,1→1}, bm_{1,2→1} ~ B_1 ~ bS_1 ≡ (bS_1 bS_2) ~ B_{1,2} ~ bm_{2,1→1},
   aS_1 ~ B_1 ~ aΔ_{1→1,2} ≡ aΔ_{1→2,1} ~ B_{1,2} ~ (aS_1 aS_2), bS_1 ~ B_1 ~ bΔ_{1→1,2} ≡ bΔ_{1→2,1} ~ B_{1,2} ~ (bS_1 bS_2)}]
```
Out[∘]= {1.29688, {True, True, True, True}}

Pairing axioms

```
In[]:= Timing[HL /@ {(bm_{1,2→1} 𝔼_{{3}→{3}}[α₃ a₃, ξ₃ x₃, 1]) ~ B_{1,3} ~ P_{1,3} ≡
      (𝔼_{{1}→{1}}[β₁ b₁, η₁ y₁, 1] 𝔼_{{2}→{2}}[β₂ b₂, η₂ y₂, 1] aΔ_{3→4,5}) ~ B_{1,4} ~ P_{1,4} ~ B_{2,5} ~ P_{2,5},
     (bΔ_{1→1,2} 𝔼_{{3}→{3}}[α₃ a₃, ξ₃ x₃, 1] 𝔼_{{4}→{4}}[α₄ a₄, ξ₄ x₄, 1]) ~ B_{1,3} ~ P_{1,3} ~ B_{2,4} ~ P_{2,4} ≡
      (𝔼_{{1}→{1}}[β₁ b₁, η₁ y₁, 1] am_{3,4→3}) ~ B_{1,3} ~ P_{1,3} }]

Out[]= {0.625, {True, True}}
```

```
In[]:= Timing[HL /@ {((bS₁ 𝔼_{{2}→{2}}[α₂ a₂, ξ₂ x₂, 1]) // P_{1,2}) ≡ ((𝔼_{{1}→{1}}[β₁ b₁, η₁ y₁, 1] aS₂) // P_{1,2}),
     (overline(bS)₁ 𝔼_{{2}→{2}}[α₂ a₂, ξ₂ x₂, 1]) ~ B_{1,2} ~ P_{1,2} ≡ (𝔼_{{1}→{1}}[β₁ b₁, η₁ y₁, 1] overline(aS)₂) ~ B_{1,2} ~ P_{1,2}}]

Out[]= {0.421875, {True, True}}
```

Tests for the double.

Check the double formulas on the generators agree with SL2Portfolio.pdf:

```
In[]:= Timing@{{
      "[a,y]" →
       ((𝔼_{{}→{1,2}}[0, 0, y₂ a₁] ~ B_{1,2} ~ dm_{1,2→1}) [3] - (𝔼_{{}→{1,2}}[0, 0, y₁ a₂] ~ B_{1,2} ~ dm_{1,2→1}) [3]),
      "[b,x]"  → ((𝔼_{{}→{1,2}}[0, 0, x₂ b₁] ~ B_{1,2} ~ dm_{1,2→1}) [3] -
        (𝔼_{{}→{1,2}}[0, 0, x₁ b₂] ~ B_{1,2} ~ dm_{1,2→1}) [3]),
      "xy-qyx" → ((𝔼_{{}→{1,2}}[0, 0, x₁ y₂] ~ B_{1,2} ~ dm_{1,2→1}) [3] -
        (1 + ϵ) (𝔼_{{}→{1,2}}[0, 0, y₁ x₂] ~ B_{1,2} ~ dm_{1,2→1}) [3])
     } /. {z_1 → z} // Expand // Factor,
    {
     "Δ(a)" → ((𝔼_{{}→{1}}[0, 0, a₁] ~ B₁ ~ dΔ_{1→1,2}) [3]),
     "Δ(x)" → ((𝔼_{{}→{1}}[0, 0, x₁] ~ B₁ ~ dΔ_{1→1,2}) [3]),
     "Δ(b)" → ((𝔼_{{}→{1}}[0, 0, b₁] ~ B₁ ~ dΔ_{1→1,2}) [3]),
     "Δ(y)" → ((𝔼_{{}→{1}}[0, 0, y₁] ~ B₁ ~ dΔ_{1→1,2}) [3])
    } // Simplify,
    {
      "S(a)" → ((𝔼_{{}→{1}}[0, 0, a₁] ~ B₁ ~ dS₁) [3]),
      "S(x)" → ((𝔼_{{}→{1}}[0, 0, x₁] ~ B₁ ~ dS₁) [3]),
      "S(b)" → ((𝔼_{{}→{1}}[0, 0, b₁] ~ B₁ ~ dS₁) [3]),
      "S(y)" → ((𝔼_{{}→{1}}[0, 0, y₁] ~ B₁ ~ dS₁) [3])
     } /. {z_1 → z} // Simplify
   }
```

Out[]= $\{10.6406, \{\{[a,y] \to -y\gamma + O[\epsilon]^3,\ [b,x] \to x\epsilon + O[\epsilon]^3,$

$xy\text{-}qyx \to \frac{1-B}{\hbar} + (a B - x y + x y \gamma \hbar)\,\epsilon + \left(-\frac{1}{2} a^2 B \hbar + \frac{1}{2} x y \gamma^2 \hbar^2\right) \epsilon^2 + O[\epsilon]^3\},$

$\{\Delta(a) \to (a_1 + a_2) + O[\epsilon]^3,\ \Delta(x) \to (x_1 + x_2) - \hbar a_1 x_2 \epsilon + \frac{1}{2} \hbar^2 a_1^2 x_2 \epsilon^2 + O[\epsilon]^3,$

$\Delta(b) \to (b_1 + b_2) + O[\epsilon]^3,\ \Delta(y) \to (y_1 + B_1 y_2) + O[\epsilon]^3\},$

$\{S(a) \to -a + O[\epsilon]^3,\ S(x) \to -x - a x \hbar \epsilon - \frac{1}{2} (a^2 x \hbar^2)\, \epsilon^2 + O[\epsilon]^3,$

$S(b) \to -b + O[\epsilon]^3,\ S(y) \to -\frac{y}{B} + \frac{y \gamma \hbar \epsilon}{B} - \frac{(y \gamma^2 \hbar^2)\, \epsilon^2}{2 B} + O[\epsilon]^3\}\}\}$

(co)-associativity

In[∘]:= $\texttt{Timing}[\texttt{HL /@} \{(d\Delta_{1\to1,2} \texttt{ // } d\Delta_{2\to2,3}) \equiv (d\Delta_{1\to1,3} \texttt{ // } d\Delta_{1\to1,2}),\ (dm_{1,2\to1} \texttt{ // } dm_{1,3\to1}) \equiv (dm_{2,3\to2} \texttt{ // } dm_{1,2\text{->}1})\}]$

Out[∘]= {6.75, {True, True}}

Δ is an algebra morphism

In[∘]:= $\texttt{Timing@HL}[dm_{1,2\to1} \sim B_1 \sim d\Delta_{1\to1,2} \equiv (d\Delta_{1\to1,3}\ d\Delta_{2\to2,4}) \sim B_{1,2,3,4} \sim (dm_{3,4\to2}\ dm_{1,2\to1})]$

Out[∘]= {7.96875, True}

S_2 inverts R, but not S_1:

In[∘]:= $\texttt{Timing@}\{R_{1,2} \sim B_1 \sim dS_1 \equiv \overline{R}_{1,2},\ \texttt{HL}[R_{1,2} \sim B_2 \sim dS_2 \equiv \overline{R}_{1,2}]\}$

Out[∘]= $\left\{1.0625, \left\{\frac{1}{4 B_1^3}\left(4 \gamma \epsilon \hbar^2 B_1^2 x_2 y_1 - 2 \gamma^2 \epsilon^2 \hbar^3 B_1^2 x_2 y_1 + 4 \gamma \epsilon^2 \hbar^3 a_2 B_1^2 x_2 y_1 + 8 \gamma^2 \epsilon^2 \hbar^4 B_1 x_2^2 y_1^2 - 4 \gamma \epsilon^2 \hbar^4 a_2 B_1 x_2^2 y_1^2 - 3 \gamma^2 \epsilon^2 \hbar^5 x_2^3 y_1^3\right) = 0, \text{True}\right\}\right\}$

S is convolution inverse of id

In[∘]:= $\texttt{Timing}[\texttt{HL}[\# \equiv \mathbb{E}_{\{1\}\to\{1\}}[0, 0, 1]]\ \texttt{\& /@}\ \{(d\Delta_{1\to1,2} \sim B_1 \sim dS_1) \sim B_{1,2} \sim dm_{1,2\to1},\ (d\Delta_{1\to1,2} \sim B_2 \sim dS_2) \texttt{ // } dm_{1,2\to1}\}]$

Out[∘]= {11.8125, {True, True}}

S is a (co)-algebra anti-morphism

In[∘]:= $\texttt{Timing}[\texttt{HL /@ Expand /@}\ \{dm_{1,2\to1} \sim B_1 \sim dS_1 \equiv (dS_1\ dS_2) \sim B_{1,2} \sim dm_{2,1\to1},\ dS_1 \sim B_1 \sim d\Delta_{1\to1,2} \equiv d\Delta_{1\to2,1} \sim B_{1,2} \sim (dS_1\ dS_2)\}]$

Out[∘]= {19., {True, True}}

Quasi-triangular axiom 1:

In[∘]:= $\texttt{Timing@HL}[R_{1,2} \sim B_1 \sim d\Delta_{1\text{->}1,3} \equiv (R_{1,4}\ R_{3,2}) \sim B_{2,4} \sim dm_{2,4\to2}]$

Out[∘]= {0.71875, True}

Quasi-triangular axiom 2:

In[∘]:= $\texttt{Timing@HL}[((d\Delta_{1\text{->}1,2}\ R_{3,4}) \sim B_{1,2,3,4} \sim (dm_{1,3\to1}\ dm_{2,4\to2})) \equiv ((d\Delta_{1\to2,1}\ R_{3,4}) \sim B_{1,2,3,4} \sim (dm_{3,1\to1}\ dm_{4,2\to2}))]$

Out[∘]= {6.125, True}

The Drinfel'd element inverse property, $(u_1\ \overline{u}_2) \sim B_{1,2} \sim \mathrm{dm}_{1,2\to1} \equiv \mathbb{E}[0, 0, 1]$:

In[∘]:= $\texttt{Timing@HL}[((R_{1,2} \sim B_1 \sim dS_1 \sim B_{1,2} \sim dm_{2,1\to i})\ (R_{1,2} \sim B_2 \sim dS_2 \sim B_2 \sim dS_2 \sim B_{1,2} \sim dm_{2,1\to j})) \sim B_{i,j} \sim dm_{i,j\to i} \equiv \mathbb{E}_{\{\}\to\{i\}}[0, 0, 1]]$

Out[∘]= {3.35938, True}

The ribbon element v satisfies $v^2 = S(u)\,u$. The spinner C=uv^{-1}. It is convenient to compute $z = S(u)\,u^{-1}$ which is something easy.

In[∘]:=
```
Timing@Block[{$k = 2},
  (((R1,2 ~ B1 ~ dS1 ~ B1,2 ~ dm2,1→i) ~ Bi ~ dSi) (R1,2 ~ B2 ~ dS2 ~ B2 ~ dS2 ~ B1,2 ~ dm2,1→j)) ~ Bi,j ~ dmi,j→i]
```
Out[∘]= $\left\{4.20313, \mathbb{E}_{\{\}\to\{i\}}\left[0, 0, \frac{1}{B_i} + \frac{\hbar a_i \epsilon}{B_i} + \frac{\hbar^2 a_i^2 \epsilon^2}{2 B_i} + O[\epsilon]^3\right]\right\}$

In[∘]:=
```
Timing@Block[{$k = 2}, HL /@ {(Ci C̄j) ~ Bi,j ~ dmi,j→i ≡ E{}→{i}[0, 0, 1], (C̄i Cj) ~ Bi,j ~ dmi,j→i ≡
    (((R1,2 ~ B1 ~ dS1 ~ B1,2 ~ dm2,1→i) ~ Bi ~ dSi) (R1,2 ~ B2 ~ dS2 ~ B2 ~ dS2 ~ B1,2 ~ dm2,1→j)) ~ Bi,j ~ dmi,j→i}]
```
Out[∘]= {4.76563, {True, True}}

Reidemeister 2:

In[∘]:=
```
Timing[HL[# ≡ E{}→{1,2}[0, 0, 1]] & /@
  {(R̄1,2 R3,4) ~ B1,2,3,4 ~ (dm1,3→1 dm2,4→2), (R1,2 R̄3,4) ~ B1,2,3,4 ~ (dm1,3→1 dm2,4→2)}]
```
Out[∘]= {4.5, {True, True}}

Cyclic Reidemeister 2:

In[∘]:=
```
Timing@HL[(R1,4 R̄5,2 C̄3) ~ B2,4 ~ dm2,4→2 ~ B1,3 ~ dm1,3→1 ~ B1,5 ~ dm1,5→1 ≡ C̄1 E{}→{2}[0, 0, 1]]
```
Out[∘]= {2.625, True}

Reidemeister 3:

In[∘]:=
```
Timing@HL[((R1,2 R4,3 R5,6) ~ B1,4 ~ dm1,4→1 ~ B2,5 ~ dm2,5→2 ~ B3,6 ~ dm3,6→3) ≡
  ((R1,6 R2,3 R4,5) ~ B1,4 ~ dm1,4→1 ~ B2,5 ~ dm2,5→2 ~ B3,6 ~ dm3,6→3)]
```
Out[∘]= {5.07813, True}

Relations between the four kinks:

In[∘]:=
```
Timing[HL /@ {Kinki ≡ (R3,1 C2) ~ B1,2 ~ dm1,2→1 ~ B1,3 ~ dm1,3→i,
   K̄inkj ≡ (R̄3,1 C̄2) ~ B1,2 ~ dm1,2→1 ~ B1,3 ~ dm1,3→j, (Kinki K̄inkj) ~ Bi,j ~ dmi,j→1 ≡ E{}→{1}[0, 0, 1]}]
```
Out[∘]= {5.0625, {True, True, True}}

The Trefoil

In[∘]:=
```
Timing@Block[{$k = 1},
  Z = R1,5 R6,2 R3,7 C̄4 K̄ink8 K̄ink9 K̄ink10;
  Do[Z = Z ~ B1,r ~ dm1,r→1, {r, 2, 10}];
  {Simplify /@ Z, Simplify /@ (Z ~ B1 ~ b2t1 /. T1 → T)}]
```
Out[∘]= $\Big\{9.875, \Big\{\mathbb{E}_{\{\}\to\{1\}}\Big[0, 0, \frac{B_1}{1 - B_1 + B_1^2} - \left(\hbar B_1 \left(-a_1 \left(-1 + B_1 - B_1^3 + B_1^4\right) + \gamma \left(B_1 - 2 B_1^2 - 2 B_1^4 + 2 \hbar x_1 y_1 + B_1^3 \left(3 + 2 \hbar x_1 y_1\right)\right)\right) \epsilon\right) \Big/ \left(1 - B_1 + B_1^2\right)^3 + O[\epsilon]^2\Big], \mathbb{E}_{\{\}\to\{1\}}\Big[0, 0, \frac{T}{1 - T + T^2} + \left(T \hbar \left(T \left(-1 + 2 T - 3 T^2 + 2 T^3\right) \gamma + 2 \left(-1 + T - T^3 + T^4\right) a_1 - 2 \left(1 + T^3\right) \gamma \hbar x_1 y_1\right) \epsilon\right) \Big/ \left(1 - T + T^2\right)^3 + O[\epsilon]^2\Big]\Big\}\Big\}$

Program

In[∘]:=
```
Define[kR_{i,j} = R_{i,j} // (b2t_i b2t_j) /. t_{i|j} → t,
  kR̄_{i,j} = R̄_{i,j} // (b2t_i b2t_j) /. {t_{i|j} → t, T_{i|j} → T},
  km_{i,j→k} = (t2b_i t2b_j) // dm_{i,j→k} // b2t_k /. {t_k → t, T_k → T, τ_{i|j} → 0},
  kC_i = C_i // b2t_i /. T_i → T,
  kC̄_i = C̄_i // b2t_i /. T_i → T,
  kKink_i = Kink_i // b2t_i /. {t_i → t, T_i → T},
  kKink̄_i = Kink̄_i // b2t_i /. {t_i → t, T_i → T}]
```

In[∘]:=
```
Timing@Block[{$k = 1},
  Z = kR_{1,5} kR_{6,2} kR_{3,7} kC̄_4 kKink̄_8 kKink̄_9 kKink̄_10;
  Do[Z = Z~B_{1,r}~km_{1,r→1}, {r, 2, 10}];
  Simplify /@ Z]
```

Out[∘]= $\{5.35938, \mathbb{E}_{\{\}\to\{1\}}[0, 0, \frac{T}{1 - T + T^2} + (T \hbar (T (-1 + 2 T - 3 T^2 + 2 T^3) \gamma + 2 (-1 + T - T^3 + T^4) a_1 - 2 (1 + T^3) \gamma \hbar x_1 y_1) \epsilon) / (1 - T + T^2)^3 + O[\epsilon]^2]\}$

RVK, rot, Z from 2016-09/OneSmidgen.nb. See also local version in this folder.

Program

Some details of the code below are at http://drorbn.net/bbs/show?shot=Dror-160920-151350.jpg.

Program

In[∘]:=
```
RVK::usage =
  "RVK[xs, rots] represents a Rotational Virtual Knot with a list of n Xp/Xm crossings
    xs and a length 2n list of rotation numbers rots. Crossing
    sites are indexed 1 through 2n, and rots⟦k⟧ is the rotation
    between site k-1 and site k. RVK is also a casting operator
    converting to the RVK presentation from other knot presentations.";
```

Program

In[∘]:=
```
RVK[pd_PD] := PP_RVK@Module[{n, xs, x, rots, front = {0}, k},
    n = Length@pd; rots = Table[0, {2 n}];
    xs = Cases[pd, x_X :> { Xp[x⟦4⟧, x⟦1⟧]  PositiveQ@x
                           Xm[x⟦2⟧, x⟦1⟧]  True        ];
    For[k = 0, k < 2 n, ++k, If[k == 0 ∨ FreeQ[front, -k],
      front = Flatten[front /. k → (xs /. {
              Xp[k + 1, l_] | Xm[l_, k + 1] :> {l, k + 1, 1 - l},
              Xp[l_, k + 1] | Xm[k + 1, l_] :> (++rots⟦l⟧; {1 - l, k + 1, l})
            })],
      Cases[front, k | -k] /. {k, -k} :> --rots⟦k + 1⟧;
     ]];
    RVK[xs, rots] ];
RVK[K_] := RVK[PD[K]];
```

In[∘]:=
```
xs = Cases[pd, x_X :> If[PositiveQ@x, Xp[x⟦4⟧, x⟦1⟧], Xm[x⟦2⟧, x⟦1⟧]]];
```

```
In[]:= RVK[Knot[10, 100]]

KnotTheory: Loading precomputed data in PD4Knots`.

Out[]= RVK[{Xp[1, 6], Xp[5, 18], Xm[13, 20], Xm[7, 14], Xm[3, 10], Xm[9, 16], Xm[11, 4], Xm[15, 8],
   Xm[19, 12], Xp[17, 2]}, {0, 0, 0, 0, 0, -1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0}]
```

Program

```
In[]:= rot[i_, 0] := 𝔼_{}→{i}[0, 0, 1];
rot[i_, n_] := Module[{j},
   rot[i, n] = If[n > 0, rot[i, n - 1] kC_j, rot[i, n + 1] kC̄_j] // km_{i,j→i}];
```

Program

```
In[]:= Z[K_] := Z[RVK@K];
Z[rvk_RVK] := (*Z[rvk] =*)
  PP"Z"@Module[{todo, n, rots, ζ, done, st, cx, ζ1, i, j, k, k1, k2, k3},
   {todo, rots} = List @@ rvk;
   AppendTo[rots, 0];
   n = Length[todo];
   ζ = 𝔼_{}→{0}[0, 0, 1];
   done = {0};
   st = Range[0, 2 n + 1];
   While[Echo@todo =!= {},
    {cx} = MaximalBy[todo, Length[done ⋂ {#[[1]], #[[2]], #[[1]] - 1, #[[2]] - 1}] &, 1];
    {i, j} = List @@ cx;
    ζ1 = Switch[Head[cx],
      Xp, (kR_{i,j} kKink̄_k) // km_{j,k→j},
      Xm, (kR̄_{i,j} kKink_k) // km_{j,k→j}
     ];
    ζ1 = (rot[k, rots[[i]]] ζ1) // km_{k,i→i}; rots[[i]] = 0;
    ζ1 = (ζ1 rot[k, rots[[i + 1]]]) // km_{i,k→i}; rots[[i + 1]] = 0;
    ζ1 = (rot[k, rots[[j]]] ζ1) // km_{k,j→j}; rots[[j]] = 0;
    ζ1 = (ζ1 rot[k, rots[[j + 1]]]) // km_{j,k→j}; rots[[j + 1]] = 0;
    ζ *= ζ1;
    If[MemberQ[done, i], ζ = ζ // km_{i,i+1→i}; st = st /. st[[i + 2]] → st[[i + 1]]];
    If[MemberQ[done, i - 1], ζ = ζ // km_{st[[i]],i→st[[i]]}; st = st /. st[[i + 1]] → st[[i]]];
    If[MemberQ[done, j], ζ = ζ // km_{j,j+1→j}; st = st /. st[[j + 2]] → st[[j + 1]]];
    If[MemberQ[done, j - 1], ζ = ζ // km_{st[[j]],j→st[[j]]}; st = st /. st[[j + 1]] → st[[j]]];
    done = done ⋃ {i - 1, i, j - 1, j};
    todo = DeleteCases[todo, cx]
   ];
   Simplify /@ (ζ /. {x_0 → x, y_0 → y, a_0 → a})
  ]
```

Knot

In[∘]:=
```
$k = 1; Timing@Z@Knot[10, 100]
```

Knot

Out[∘]=

$$\{174.172, \mathbb{E}_{\{\}\to\{0\}}[0, 0, T^4 / (1 - 4 T + 9 T^2 - 12 T^3 + 13 T^4 - 12 T^5 + 9 T^6 - 4 T^7 + T^8) + (T^4 \hbar (4 a (-2 + 14 T - 51 T^2 + 120 T^3 - 203 T^4 + 258 T^5 - 246 T^6 + 152 T^7 - 152 T^9 + 246 T^{10} - 258 T^{11} + 203 T^{12} - 120 T^{13} + 51 T^{14} - 14 T^{15} + 2 T^{16}) + \gamma (-6 + 2 T^{16} - 8 x y \hbar - 440 T^9 (-1 + x y \hbar) - 4 T^{15} (3 + 2 x y \hbar) + 8 T^8 (-97 + 21 x y \hbar) + 8 T^7 (131 + 21 x y \hbar) - 20 T^6 (57 + 22 x y \hbar) + T^{14} (37 + 48 x y \hbar) + T (44 + 48 x y \hbar) - 8 T^{11} (2 + 61 x y \hbar) + 8 T^5 (127 + 68 x y \hbar) - 2 T^{13} (35 + 78 x y \hbar) + 4 T^{10} (-39 + 136 x y \hbar) - T^2 (167 + 156 x y \hbar) + T^{12} (79 + 324 x y \hbar) + T^3 (410 + 324 x y \hbar) - T^4 (733 + 488 x y \hbar))) \epsilon) / (1 - 4 T + 9 T^2 - 12 T^3 + 13 T^4 - 12 T^5 + 9 T^6 - 4 T^7 + T^8)^3 + O[\epsilon]^2]\}$$

In[∘]:=
```
EndProfile[];
```

Profile

In[∘]:=
```
PrintProfile[]
```

Profile

Out[∘]=
```
ProfileRoot is root. Profiled time: 300.249
  (      1)    0.155/ 174.156 above Z
  (    157)    0.593/ 106.422 above B
  (     37)    0.111/  19.531 above Boot
  (    147)    0.079/   0.125 above CF
  (      2)    0.015/   0.015 above RVK
CF: called 13049 times, time in 72.658/199.97
  (   2694)   30.392/ 138.454 under LZip
  (    147)    0.079/   0.125 under ProfileRoot
  (  10208)   42.187/  61.391 under QZip
  (  39202)   33.196/ 127.312 above CCF
LZip: called 294 times, time in 63.015/211.391
  (    294)   63.015/ 211.391 under B
  (   2694)   30.392/ 138.454 above CF
  (    294)    1.942/   9.922 above Zip
Together: called 160474 times, time in 55.54/109.531
  ( 160474)   55.540/ 109.531 under CCF
  ( 160474)   18.779/  53.991 above Exp
CCF: called 160474 times, time in 52.993/162.524
  (  39202)   33.196/ 127.312 under CF
  ( 121272)   19.797/  35.212 under Exp
  ( 160474)   55.540/ 109.531 above Together
QZip: called 294 times, time in 19.375/87.388
  (    294)   19.375/  87.388 under B
  (  10208)   42.187/  61.391 above CF
  (    294)    1.560/   6.622 above Zip
Exp: called 160474 times, time in 18.779/53.991
  ( 160474)   18.779/  53.991 under Together
  ( 121272)   19.797/  35.212 above CCF
Zip: called 2675 times, time in 13.904/54.531
  (    294)    1.942/   9.922 under LZip
  (    294)    1.560/   6.622 under QZip
  (   2087)   10.402/  37.987 under Zip
  (   2675)    2.640/   2.640 above Collect
```

```
  (  2087)   10.402/  37.987 above Zip
Collect: called 2675 times, time in 2.64/2.64
  (  2675)    2.640/   2.640 under Zip
B: called 294 times, time in 0.845/299.624
  (    72)    0.142/ 173.798 under Z
  (    65)    0.110/  19.404 under Boot
  (   157)    0.593/ 106.422 under ProfileRoot
  (   294)   63.015/ 211.391 above LZip
  (   294)   19.375/  87.388 above QZip
Boot: called 59 times, time in 0.33/27.608
  (     3)       0/   0.203 under Z
  (    19)    0.219/   7.874 under Boot
  (    37)    0.111/  19.531 under ProfileRoot
  (    65)    0.110/  19.404 above B
  (    19)    0.219/   7.874 above Boot
Z: called 1 times, time in 0.155/174.156
  (     1)    0.155/ 174.156 under ProfileRoot
  (    72)    0.142/ 173.798 above B
  (     3)       0/   0.203 above Boot
RVK: called 2 times, time in 0.015/0.015
  (     2)    0.015/   0.015 under ProfileRoot
```